[Question]
  [
The goal of this challenge is to check and extend the OEIS sequence [A334248](http://oeis.org/A334248): Number of distinct acyclic orientations of the edges of an n-dimensional cube.
Take an n-dimensional cube (if n=1, this is a line; if n=2, a square; if n=3, a cube; if n=4, a hypercube/tesseract; etc), and give a direction to all of its edges so that no cycles are formed. The terms of the sequence are the number of (rotationally and reflectively) distinct orientations that can be obtained.
## Example
For n=2, there are 14 ways to assign directions to the edges of a square with no cycles formed:
![Acyclic orientations of a cube](https://upload.wikimedia.org/wikipedia/commons/thumb/0/09/Acyclic_orientations_of_C4.svg/477px-Acyclic_orientations_of_C4.svg.png)
However, only three of these are distinct if they are rotated and reflected:
![Three distinct acyclic orientations](https://i.stack.imgur.com/6rTCG.png)
So when n=2, the answer is 3.
---
The winner of this [fastest-code](/questions/tagged/fastest-code "show questions tagged 'fastest-code'") challenge will be the program that calculates the most terms when run on my computer (Ubuntu, 64GB RAM) in one hour. If there is a tie, the program that computes the terms in the shortest time will be the winner.
  
[Answer]
# [Wolfram Language (Mathematica)](https://www.wolfram.com/wolframscript/), n = 4, <1 second on my computer
```
(* Elements of the hyperoctahedral group,represented by matrices *)
HyperoctahedralGroup[n_] := 
  Join @@ Table[
    Permute[IdentityMatrix@n, perm]*sign, {perm, 
     Permutations@Range@n}, {sign, Tuples[{1, -1}, n]}];
(* Action of group elements on a vertex of the hypercube,
where each vertex is represented by an integer from 1 to 2^n *)
ActOn[n_, g_, v_] := 
  FromDigits[(1 + (IntegerDigits[v - 1, 2, n]*2 - 1) . g)/2, 2] + 1;
(* Orbits of vertices under such action *)
Orbits[n_, g_] := 
  ConnectedComponents@Table[v <-> ActOn[n, g, v], {v, 2^n}];
(* Whether any orbit contains adjacent vertices *)
OrbitConatinsAdjacentVertices[edges_, orbits_] := 
  Or @@ Or @@@ Outer[SubsetQ, orbits, edges, 1];
(* Merging vertices in each orbit into a single vertex *)
MergeVertices[edges_, orbits_] := 
  Graph@DeleteDuplicates[
    Sort /@ (edges /. (Alternatives@##2 -> #1 & @@@ orbits))];
(* Number of acyclic orientations fixed by some group element *)
InvariantAcyclicOrientations[n_, edges_, g_] := 
  With[{orbits = Orbits[n, g]}, 
   If[OrbitConatinsAdjacentVertices[edges, orbits], 0, 
    Abs[ChromaticPolynomial[MergeVertices[edges, orbits], -1]]]];
(* OEIS sequence A334248 *)
A334248[n_] := 
  With[{edges = List @@@ EdgeList@HypercubeGraph@n}, 
   Sum[InvariantAcyclicOrientations[n, edges, g], {g, 
      HyperoctahedralGroup[n]}]/(2^n*n!)];
Do[Print[n -> A334248[n]], {n, 1, 4}]
```
[Try it online!](https://tio.run/##jVRfb9MwEH/vpzg0CbUlXUm3BwRsSrUOVsTooBN7sAJy02tilNjFdsKqqZ@9nOOk29AEVGqbOHeX3z@74DbDgluR8N2u24fzHAuU1oBaAT2BbLNGrRLLM1xqnkOqVbkONK41GqrDJSw2QP1aJGig3@tcPG547@qZ/B7D6xPoAHxQQkIUwTVf5MhoAeAKdVFaZNMlDRR2c@mm3UYyAJpUxH0jUrq@czcB1B1NC6FW0kRfuEwxkluq8aXX5TpHw@7CAAYhLct4G7/pdIjdOHEtjlvNA3DPVgKHCrXF20fMk3KBQedXhhoBeZK1NcLAHxpwCYKuU9Sw0qqAEKyC0TfpNKHXziSJEEBK32ovxjsqnIhUWMO6IbyA7tRPaNYqGABxGDkG/ZG76cEhpL0hrYxiqg89q5leCO@YQ1cbUcolATElIeaeM8HwdQ2OFsOZkhITInGmirWSTo3Im1PB28EpNNiphZDHpHEVOFqtojcZklia6G9AufmQKGm5kAb48gdPaN49qhYDvZO8k2bcVHxtChguUzQEr55k9hhn2kWm/qU/yopm83Jh0H5uSwOoWwMIG1yXqFMh0/t3U@xqAz1IskqR44ZKcmxNJXiuDf8F573m6yyaUHgsTihrtHcsVdfRnCttYRhBt@6F4SF0xzkBdoQrNNHBARl5CgchPK/Z@NG9XgP7U1ksSE2ykiebhCZTgSCFfNRhJW592owq8HGGHfqprLgWXNqxb5496K1tbwnd238jbMbuPAo4gTYiVBJv/W6brth/mNaKRAl52ezS8cKws4wi7k6XK5VvpCoEz9kTGj/oHoQxfZpgn0/nYPBniTJBGB8dHY@OX9Ubyl8@OFc8Dy/6CXwUxtbyntOCu4ku2t3szZMNuXlZsL@rtk9W6sKfticQPH3O0Ukz7NL26MtntaUTxa40hY1J5/oeduxm0Wja3cfbeLf7DQ "Wolfram Language (Mathematica) – Try It Online")
Not really optimized. Takes less than 1 seconds on my computer to calculate the first four terms, but takes forever for the fifth term. It seems that Mathematica cannot compute the chromatic polynomial of `HypercubeGraph[5]` in a reasonable time.
Based on Peter Kagey's deleted answer (which did not consider rotations and reflections), and uses Burnside's lemma.
]